Разбор задач районного тура Всероссийской олимпиады школьников по программированию.

Задача A. Камень-ножницы-бумага

Если слегка переформулировать условие, задача сводится к рассмотрению нескольких случаев:

· если найдется игрок победивший всех, ему дается 2 очка;
· если найдется игрок проигравший всем, остальным дается по 1 очку;
· в других случаях очков никому не дается.

Одно из решений предполагает рассмотрение всех вариантов “вручную”, множеством конструкций if .. then. Но в таком решении присутствует существенный минус – каждый из случаев уникален, и ошибку, совершенную из-за невнимательности или опечатки не так легко обнаружить.
Поскольку вариантов выигравшего или проигравшего игрока – 3, а задача проверки в этих случаях одинаковая, логично хранить номера жестов, показанных игроками в массивах, а процедуру проверки реализовывать при помощи циклов. 
Решение Бочкарева Михаила использует эту технологию, а также довольно оригинальную и простую проверку на выигрыш. Ниже оно проводится с комментариями.
	var   n,i,j,k   :integer ;

          a,b  :array[1..3] of integer;
// a[i] – жест i-го игрока
// b[i] – сумма очков i-го игрока
           c  :array[1..3,1..2] of byte;
// В элементе c[i,1] хранится количество игроков в текущей партии, у которых выйграл i-ый игрок

// В элементе c[i,2] которым i-ый проиграл
begin
 reset(input,'input.txt');

 rewrite(output,'output.txt');

 read(n);

 for k:=1 to n do // цикл по партиям
  begin
   // ввод данных партии
   read(a[1],a[2],a[3]);

   for i:=1 to 3 do

   for j:=1 to 2 do

    c[i,j]:=0;

   for i:=1 to 3 do

   for j:=1 to 3 do // перебираем пары игроков
    if i<>j then

     begin
      // увеличиваем в случаях выигрыша, или проигрыша соответствующие ячейки массива
      if (a[i]*2=a[j]) or (a[i] div 4=a[j]) then inc(c[i,1]); // i выиграл у j
      if (a[j]*2=a[i]) or (a[j] div 4=a[i]) then inc(c[i,2]); // j выиграл у i
     end;

   for i:=1 to 3 do

    begin

     if c[i,1]=2 then b[i]:=b[i]+2; // i выиграл у двоих других
     if (c[i,1]=1) and(c[i,2]=0) then b[i]:=b[i]+1; // 2-ой случай (см.выше)
    end;

  end;

 // вывод
 write(b[1],' ',b[2],' ',b[3]);

end.


Задача B. Алиса в Зазеркалье 
После прочтения задача может показаться сложной, или не совсем понятной из-за тестового примера.

Основная сложность в том, что, например, при движении стрелки минут, часовая тоже медленно вращается.

Однако немного подумав, на какие позиции будут показывать стрелки при зеркальном отражении, вырисовывается несколько случаев.

Пусть H, M, S – исходное положение стрелок часов, минут и секунд.
Если секундная и минутная стрелки стоят на нуле, то при отражении часовая будет указывать на позицию 12-H, а т.к. 0 и 12 часов – одно и тоже, то новое значение часов будет - (12-H) mod 12. В противном случае значение часов будет на один час меньше.
Аналогично рассматриваются случаи для минутной стрелки. 

Но самое простое решение данной задачи предложил Детянцев Павел.
Будем считать, что время на часах – до полудня.

Идея заключается в том, что при обратном вращении стрелок, отрезок времени, между моментом времени на часах и полудня, совпадает с отрезком времени на нормальных часах между началом суток и текущим временем на часах.

Тогда нам нужно перевести текущее время в секунды, вычесть его из длительности половины суток, и перевести обратно.
	var T,H,M,S:longint;

begin

assign(input,'input.txt');

assign(output,'output.txt');

Reset(Input);

Rewrite(output);

Read(H,M,S); // ввод
T:=12*60*60-(H*60*60+M*60+S); // переводим время, и вычитаем его из длительности полусуток
Write((T div (60*60)) mod 12,' ', (T div 60) mod 60,' ', T mod 60); // вывод
Close(input);

Close(output);

end.


Задача C. Волшебная последовательность
Разные вариации задачи достаточно распространены, особенно в математике.
Но эта задача, из-за небольших ограничений подразумевала решение в “лоб”. Т.е. генерацию последовательности, а потом поиск первого вхождения.

Возьмем начало последовательности, в которой мы ищем, до позиции с 10-ю нулями. Несложно заметить, что любая подстрока, состоящая не более чем из 10 символов, входящая  в полную последовательность, входит и в выбранное нами начало.
Выпишем выбранную часть последовательности, и воспользуемся функцией Pos(a,b) – возвращающей позицию первого вхождения строки a в b, и получим самое короткое решение, предложенное Белоковским Семеном.

Функция Pos возвращает 0, если вхождения нет.
	var s,s1:string;

    l:integer;

begin

  assign(input,'input.txt');

  assign(output,'output.txt');

  reset(input);

  rewrite(output);

  readln(s); // ввод
  s1:='110100100010000100000100000010000000100000000100000000010000000000'; // начало посл-сти
  l:=pos(s,s1); // находим вхождение
  write(l); // вывод
  close(input);

  close(output);

end.


Задача D. Маленькая страна
Задача на знание вычислительной геометрии.

Решения могли опираться на разные комбинации суммирования/вычитания площадей треугольников. Другие решения могли использовать тригонометрию, и вычисления в вещественных числах с плавающей точкой.

Но самое простое решение использует известную формулу ориентированной площади произвольного, не обязательно выпуклового, многоугольника.

Например, одно из решений, использующих формулу – решение Комарова Андрея.

Идея формулы заключается в суммировании векторных произведений. Итоговая сумма равна удвоенной площади многоугольника, со знаком +, если порядок вершин в многоугольнике против часовой стрелки, или со знаком -, в противном случае.
	type tt=record x,y:integer; end; // используем тип запись для хранения координат вершины
var a:array[1..1000]of TT; // массив вершин
i,n,s:integer;

begin

  reset(input,'input.txt');

  rewrite(output,'output.txt');

  s:=0;

  read(n);

  for i:=1 to n do read(a[i].x,a[i].y); // ввод
  for i:=1 to n-1 do s:=s+a[i].x*a[i+1].y-a[i+1].x*a[i].y; // считаем ориентированную
  s:=s+a[n].x*a[1].y-a[1].x*a[n].y; // площадь произвольного многоугольника
  write(abs(s div 2)); // выводим ответ
end.


